

## CprE 288 – Introduction to Embedded Systems ATmega128 Assembly Programming: Moving Data & Control Flow

Instructors:

Dr. Phillip Jones (Sections F, G, J)

Dr. Zhao Zhang (Sections A, B, C, D, E)

1

## Major Classes of Assembly Instructions

- Data Movement
  - Move data between registers
  - Move data in & out of SRAM
  - Different addressing modes
- Logic & Arithmetic
  - Addition, subtraction, etc.
  - AND, OR, bit shift, etc.
- Control Flow
  - Control which sections of code should be executed (e.g. In C “IF”, “CASE”, “WHILE”, etc.
  - Typically the result of Logic & Arithmetic instructions help decided what path to take through the code.

CprE 288, ISU, Fall 2011

2

## Major Classes of Assembly Instructions

- Data Movement
  - Move data between registers
  - Move data in & out of SRAM
  - Different addressing modes
- Logic & Arithmetic
  - Addition, subtraction, etc.
  - AND, OR, bit shift, etc.
- Control Flow
  - Control which sections of code should be executed (e.g. In C “IF”, “CASE”, “WHILE”, etc.
  - Typically the result of Logic & Arithmetic instructions help decided what path to take through the code.

CprE 288, ISU, Fall 2011

3

## Instructions to move data: Summary


- LDI Rd, K **Load Immediate**  $Rd \leftarrow K$  1 clk
- MOV Rd, Rr **Move Between Registers**  $Rd \leftarrow Rr$  1 clk
- LDS Rd, (k) **Load Direct**  $Rd \leftarrow (k)$  2 clks
  - Note: There is a ST version for each LD (except for LDI)
- LD Rd, Y **Load Indirect**  $Rd \leftarrow (X)$  2 clks
- LD Rd, Y+ **Load Indirect & Post-Inc.**  $Rd \leftarrow (X), X \leftarrow X + 1$  2clks
- LDD Rd, Y+q **Load Indirect + offset.**  $Rd \leftarrow (X+q)$  2 clks

CprE 288, ISU, Fall 2011

4

## LDI & MOV

- LDI Rd, K **Load Immediate**  $Rd \leftarrow K$  1 clk
- MOV Rd, Rr **Move Between Registers**  $Rd \leftarrow Rr$  1 clk
- Only need one clock cycle to execute
  - All parameters needed for execution available to ALU



- Reference: (doc0856) [AVR Instruction Set Manual](#)
  - Pages 3-7

CprE 288, ISU, Fall 2011

5

## Load Immediate

```
char a;
...
a = 0x10;
```

```
ldi    r24, 0x10    ; Load imm 10
sts    a, r24        ; Store to a
```

CprE 288, ISU, Fall 2011

6

## Load Immediate

LDI: Load an 8-bit constant (limited to R16-R31)

Syntax: **LDI Rd, K**

Operands:  $16 \leq d \leq 31$ ,  $0 \leq K \leq 255$

Operations:  $Rd \leftarrow K$ ,  $PC \leftarrow PC+1$

Binary

Format

1110	KKKK	dddd	KKKK
------	------	------	------

Cycles: 1

See [8-bit AVR Inst. Set Page 89](#)

*Question: Why limited to R16-R31?*

CprE 288, ISU, Fall 2011

7

## Copy Register

MOV: Copy one register to another

Syntax: **MOV Rd, Rr**

Operands:  $0 \leq d \leq 31$ ,  $0 \leq r \leq 31$

Operations:  $Rd \leftarrow Rr$ ,  $PC \leftarrow PC+1$

Example:

`mov r16,r0 ; Copy r0 to r16`

CprE 288, ISU, Fall 2011

8

## Copy Register Word

MOVW: Copy one register to another

Syntax (AVR): **MOVW Rd+1:Rd, Rr+1:Rr**

Syntax (GCC): **MOVW Rd, Rr**

Operands:  $d=0,2,\dots,30$ ,  $r=0,2,\dots,30$

Operations:  $Rd+1:Rd \leftarrow Rr+1:Rr$ ,  $PC \leftarrow PC+1$

Example:

(AVR) `movw r17:16, r1:r0`

(GCC) `movw r16, r0`

CprE 288, ISU, Fall 2011

9

## Copy Register & Copy Register Word

Make R2 = 0x10

– Recall: Cannot use LDI on R2

`ldi r24, 0x10 ; r24 = 0x10`

`mov r2, r24 ; r2 = r24`

Make R5:R4 = 0x3020 using three instructions

`ldi r24, 0x20 ; r24 = 0x20`

`ldi r25, 0x30 ; r25 = 0x30`

`movw r4, r24 ; r5:r4=r25:r24`

10

## Exercise

```
int a;
...
a = 0x2030;
```

11

CprE 288, ISU, Fall 2011

## Load and Store

```
int a;
...
a = 0x2030;
```

`ldi r24, 0x30 ; r24 = 0x30`

`sts a, r24 ; Store to lower half`

`ldi r24, 0x20 ; r24 = 0x20`


`sts a+1, r24 ; Store to higher half`

12

CprE 288, ISU, Fall 2011

## LDS (Load Direct from Data Space)

- LDS Rd, (k) **Load Direct**  $Rd \leftarrow (k)$  2 clks
- Note:



CprE 288, ISU, Fall 2011

13

## Load and Store

```
char a, b;
```

```
...
```

```
a = b;
```

```
lds r24, b      ; Load from b
sts a, r24      ; Store to a
```

CprE 288, ISU, Fall 2011

14

## Load Direct

LDS: Load direct from storage space

Syntax: **LDS Rd, k**

Operands:  $0 \leq d \leq 31$ ,  $0 \leq k \leq 65,535$

Operations:  $Rd \leftarrow (k)$ ,  $PC \leftarrow PC+2$

Binary  
Format

1001	000d	dddd	0000
kkkk	kkkk	kkkk	kkkk

Cycles: 2

See [8-bit AVR Inst. Set Page 90](#)

CprE 288, ISU, Fall 2011

15

## Store Direct

STS: Store direct to storage space

Syntax: **STS k, Rr**

Operands:  $0 \leq r \leq 31$ ,  $0 \leq k \leq 65,535$

Operations:  $(k) \leftarrow Rr$ ,  $PC \leftarrow PC+2$

Binary  
Format

1001	001d	dddd	0000
kkkk	kkkk	kkkk	kkkk

Cycles: 2


See [8-bit AVR Inst. Set Page 113](#)

CprE 288, ISU, Fall 2011

16

## LD (Load Indirect from Data Space)

- LD Rd, X **Load Indirect**  $Rd \leftarrow (X)$



CprE 288, ISU, Fall 2011

17

## X, Y, Z Registers

Three **indirect address (pointer) registers**: X, Y, and Z

$X \leftrightarrow R27:R26$

$Y \leftrightarrow R29:R28$

$Z \leftrightarrow R31:R30$

Use the GPR names to manipulate the pointers

Use the X, Y, Z names to dereference

CprE 288, ISU, Fall 2011

18

### Example: Load Using a Pointer

```
char ch = *str;
```

How many loads do we have to use?

Steps:

1. Load the pointer variable str
2. Load the dereferenced variable \*str
3. Store to ch

CprE 288, ISU, Fall 2011

19

### Example: Load Using a Pointer

```
char ch = *str;
```

```
; Use the Z register (Rr1:r30)
lds r30, str      ; Load str, lo8
lds r31, str+1    ; Load str, hi8
ld  r24, Z        ; load *str
sts ch, r24       ; store to a
```

Note: Z is R31:R30

CprE 288, ISU, Fall 2011

20

### Example: Load Using a Pointer

```
int a = *pInt;
```

```
; Use the Z register (r31:r30)
lds r30, pInt     ; Load pInt
lds r31, pInt+1   ;
ld  r24, Z        ; load (*pInt)
ldd r25, Z+1      ;
sts a, r24        ; store to a
sts a+1, r25      ;
```

CprE 288, ISU, Fall 2011

21

### X, Y, Z Registers

Three formats for loading indirect using X

LD Rd, X	X: Unchanged
LD Rd, X+	X: Post increment $Rd \leftarrow (X), X \leftarrow X+1$
LD Rd, -X	X: Pre decrement $X \leftarrow X-1, Rd \leftarrow (X)$

Rd can be any of R0-R31

Latency: 2 clks

CprE 288, ISU, Fall 2011

22

### X, Y, Z Registers

Three formats for storing indirect using X

ST X, Rr	X: Unchanged
ST X+, Rr	X: Post increment $(X) \leftarrow Rr, X \leftarrow X+1$
ST -X, Rr	X: Pre decrement $X \leftarrow X-1, (X) \leftarrow Rr$

Rd can be any of R0-R31

Latency: 2 clks

CprE 288, ISU, Fall 2011

23

### X, Y, Z Registers

Four formats using for loading indirect using Y or Z (Z as example)

LD Rd, Z	Z: Unchanged
LD Rd, Z+	Z: Post increment
LD Rd, -Z	Z: Pre decrement
LDD Rd, Z+q	Z: unchanged

Rd can be any of R0-R31

q is from 0 to 63 (6-bit)

CprE 288, ISU, Fall 2011

24

## X, Y, Z Registers

Four formats for storing indirect using Y and Z (Z as example)

**ST Z, Rr**      Z: Unchanged  
**ST Z+, Rr**      Z: Post increment  
**ST -Z, Rr**      Z: Pre decrement  
**STD Z+q, Rr**    Z: unchanged

Rd can be any of R0-R31

q is from 0 to 63 (6-bit)

CprE 288, ISU, Fall 2011

25

## Example of Encoding

Syntax (using Z): **LDD Rd, Z+q**

Operands:  $0 \leq d \leq 31$ ,  $0 \leq q \leq 63$

Operations:  $Rd \leftarrow (Z+q)$ ,  $PC \leftarrow PC+1$

Binary

Format

10q0	qq0d	dddd	0qqq
------	------	------	------

Cycle: 2

See [8-bit AVR Inst. Set Page 88](#)

Note: Unique coding for X, Y, and Z


Where is the index of Z (R31:R30)?

CprE 288, ISU, Fall 2011

26

## LDD X+ (Load Indirect & post increment)

- LD Rd, X+ **Load Indirect & Post-Inc.**  $Rd \leftarrow (X)$ ,  $X \leftarrow X + 1$   
2 clks




CprE 288, ISU, Fall 2011

27

## LDD Y+q (Load Indirect with Displacement)

- LDD Rd, Y+q **Load Indirect + offset.**  $Rd \leftarrow (Y+q)$  2 clks



- What is the significance of q being 6-bits in size?

CprE 288, ISU, Fall 2011

28

## Array Access

```
extern int A[], B[];
A[0] = B[0];
```

First initialize X and Z registers: RegX=A, RegZ=B

```
ldi r26, lo8(A) ; RegX = A
ldi r27, hi8(A) ;
ldi r30, lo8(B) ; RegZ = B
ldi r31, hi8(B) ;
```

Recall, array names are address constants

Note: lo8 and hi8 are gcc assembly macros

29

## Array Access

Then, load B[0] and store to A[0]

```
ld r24, Z+ ; r25:r24 = B[0]
ld r25, Z+ ;
st X+, r24 ; A[0] = r25:r24
st X+, r25 ;
```

The whole array can be copied if the code continues

30

## Major Classes of Assembly Instructions

- **Data Movement**
  - Move data between registers
  - Move data in & out of SRAM
  - Different addressing modes
- **Logic & Arithmetic**
  - Addition, subtraction, etc.
  - AND, OR, bit shift, etc.
- **Control Flow**
  - Control which sections of code should be executed (e.g. In C “IF”, “CASE”, “WHILE”, etc.
  - Typically the result of Logic & Arithmetic instructions help decided what path to take through the code.

CprE 288, ISU, Fall 2011

31

## Arithmetic Instruction

Overview of arithmetic instructions

**Addition:** ADD, ADC, ADIW

**Subtraction:** SUB, SUBI, SBC, SBCI, SBIW

**Logic:** AND, ANDI, OR, ORI, EOR

**Complements:** COM, NEG

**Register Bit Manipulation:** SBR, CBR

**Register Manipulation:** INC, DEC, TST, CLR, SER

**Multiplication1:** MUL, MULS, MULSU

**Fractional Multiplication1:** FMUL, FMULS, FMULSU

Source: AVR Studio 4 and ATmega128: A Beginner's Guide, Page 31

CprE 288, ISU, Fall 2011

32

## Arithmetic Instruction

```
int a, b;
...
a = a + b;
```

CprE 288, ISU, Fall 2011

33

## Arithmetic Instruction

```
lds    r18, a      ; load a
lds    r19, a+1    ;
lds    r24, b      ; load b
lds    r25, b+1    ;
add     r24, r18    ; add lower half
adc     r25, r19    ; add higher half
sts     a+1, r25    ; store a.byte1
sts     a, r24      ; store a.byte0
```

CprE 288, ISU, Fall 2011

34

## Add without Carry

ADD: Add two registers without carry

Syntax: **ADD Rd, Rr**

Operands:  $0 \leq d \leq 31$ ,  $0 \leq r \leq 31$

Operations:  $Rd \leftarrow Rd + Rr$ ,  $PC \leftarrow PC + 1$

Binary  
Format

0000	11rd	dddd	rrrr
------	------	------	------

SREG

I	T	H	S	V	N	Z	C
–	–	↔	↔	↔	↔	↔	↔

CprE 288, ISU, Fall 2011

35

## Add with Carry

ADC: Add two registers **with** carry

Syntax: **ADC Rd, Rr**

Operands:  $0 \leq d \leq 31$ ,  $0 \leq r \leq 31$

Operations:  $Rd \leftarrow Rd + Rr + C$ ,  $PC \leftarrow PC + 1$

Binary  
Format

0001	11rd	dddd	rrrr
------	------	------	------

SREG

I	T	H	S	V	N	Z	C
–	–	↔	↔	↔	↔	↔	↔

CprE 288, ISU, Fall 2011

36

## Arithmetic Instruction

```
int a;
...
a -= 0x4321;
```

CprE 288, ISU, Fall 2011

37

## Arithmetic Instruction

```
lds  r24, a      ; load from a
lds  r25, a+1    ;
subi  r24, 0x21   ; sub imm 0x21
sbci  r25, 0x43   ; sub imm 0x43 with
                    ; carry
sts   a+1, r25    ; store a
sts   a, r24      ;
```

CprE 288, ISU, Fall 2011

38

## Subtract Immediate

SUBI: Subtract a register and a constant

Syntax: **SUBI Rd, K**

Operands:  $16 \leq d \leq 31$ ,  $0 \leq K \leq 255$

Operations:  $Rd \leftarrow Rd - K$ ,  $PC \leftarrow PC + 1$

Binary  
Format

0101	KKKK	dddd	KKKK
------	------	------	------

SREG

I	T	H	S	V	N	Z	C
—	—	↔	↔	↔	↔	↔	↔

CprE 288, ISU, Fall 2011

39

## Subtract Immediate with Carry

SBCI: Add two registers with carry

Syntax: **SBCI Rd, K**

Operands:  $16 \leq d \leq 31$ ,  $0 \leq K \leq 255$

Operations:  $Rd \leftarrow Rd - K - C$ ,  $PC \leftarrow PC + 1$

Binary  
Format

0100	KKKK	dddd	KKKK
------	------	------	------

SREG

I	T	H	S	V	N	Z	C
—	—	↔	↔	↔	↔	↔	↔

CprE 288, ISU, Fall 2011

40

## Arithmetic Instruction

```
char a;
...
a += 0x0F;
```

How to write the assembly code?

*Challenge: There are no “ADDI” and “ADIC”!*

CprE 288, ISU, Fall 2011

41

## Logical AND

AND: Logical AND of two registers

Syntax: **AND Rd, Rr**

SREG

I	T	H	S	V	N	Z	C
—	—	—	↔	0	↔	↔	—

Example:

and r2,r3 ; Bitwise and r2 and r3, result in r2  
ldi r16,1 ; Set bitmask 0000 0001 in r16  
and r2,r16 ; Isolate bit 0 in r2

CprE 288, ISU, Fall 2011

42

## Logical AND with Immediate

ANDI: Logical AND of a register and a constant

Syntax: **ANDI Rd, K** ( $16 \leq r \leq 31$ ,  $0 \leq K \leq 255$ )

SREG

I	T	H	S	V	N	Z	C
—	—	⇔	⇔	⇔	⇔	⇔	⇔

Example:

andi r17,\$0F ; Clear upper nibble of r17

andi r18,\$10 ; Isolate bit 4 in r18

andi r19,\$AA ; Clear odd bits of r19

CprE 288, ISU, Fall 2011

43

## Multiply Unsigned

MUL: Multiply unsigned two registers

Syntax: **MUL Rd, Rr**

Operation:  $R1:R0 \leftarrow R_d \times R_r$  (unsigned)

SREG

I	T	H	S	V	N	Z	C
—	—	—	—	—	—	⇔	⇔

Example:

mul r5,r4 ; Multiply unsigned r5 and r4

movw r4,r0 ; Copy result back in r5:r4

CprE 288, ISU, Fall 2011

44

## Major Classes of Assembly Instructions

- Data Movement
  - Move data between registers
  - Move data in & out of SRAM
  - Different addressing modes
- Logic & Arithmetic
  - Addition, subtraction, etc.
  - AND, OR, bit shift, etc.
- Control Flow
  - Control which sections of code should be executed (e.g. In C "IF", "CASE", "WHILE", etc.
  - Typically the result of Logic & Arithmetic instructions help decided what path to take through the code (i.e. they set flags)

CprE 288, ISU, Fall 2011

45

## Control Flow Instruction Examples

Set Flags	CPI Rd,K	Compare with Imm	$R_d - K$
	CP Rd,Rr	Compare	$R_d - R_r$
Branch on conditions	CPC Rd,Rr	Compare with Carry	$R_d - R_r - C$
	BREQ k	Branch if Equal	if (Z=1) then $PC \leftarrow PC + k + 1$
	BRNE k	Branch if Not Equal	if (Z=0) then $PC \leftarrow PC + k + 1$
Unconditional branch	BRIE k	Branch if Interrupt Enabled	if (I=1) then $PC \leftarrow PC + k + 1$
	RJMP k	RJMP k	$PC \leftarrow PC + k + 1$
	IJMP	Indirect Jump to (Z)	$PC \leftarrow Z$
	RCALL k	Relative Call Subroutine	$PC \leftarrow PC + k + 1$
	RET	Subroutine Return	$PC \leftarrow \text{STACK}$
Compare & Branch (skip next) on condition	RETI	Interrupt Return	$PC \leftarrow \text{STACK}$
	CPSE Rd,Rr	Compare, Skip if Equal	if (Rd = Rr) $PC \leftarrow PC + 2$
	SBRC Rr, b	Skip if Bit in Register Cleared	if (Rr(b)=0) $PC \leftarrow PC + 2$
	SBRS Rr, b	Skip if Bit in Register Set	if (Rr(b)=1) $PC \leftarrow PC + 2$

CprE 288, ISU, Fall 2011

46

## Summary of Branch Conditions

Table 7: Conditional Branch Summary

Test	Boolean	Mnemonic	Complementary	Boolean	Mnemonic	Comment
$R_d > R_r$	$Z \wedge (N \oplus V) = 0$	BRLT <sup>(1)</sup>	$R_d \leq R_r$	$Z \wedge (N \oplus V) = 1$	BRGE	Signed
$R_d \geq R_r$	$(N \oplus V) = 0$	BRGE	$R_d < R_r$	$(N \oplus V) = 1$	BRLT	Signed
$R_d = R_r$	$Z = 1$	BREQ	$R_d \neq R_r$	$Z = 0$	BRNE	Signed
$R_d < R_r$	$Z \wedge (N \oplus V) = 1$	BRLE <sup>(1)</sup>	$R_d \geq R_r$	$Z \wedge (N \oplus V) = 0$	BRLT	Signed
$R_d < R_r$	$(N \oplus V) = 1$	BRLT	$R_d \geq R_r$	$(N \oplus V) = 0$	BRGE	Signed
$R_d > R_r$	$C + Z = 0$	BRLO <sup>(1)</sup>	$R_d \leq R_r$	$C + Z = 1$	BRSH	Unsigned
$R_d \geq R_r$	$C = 0$	BRSH/BRCC	$R_d < R_r$	$C = 1$	BRLO/BRCS	Unsigned
$R_d = R_r$	$Z = 1$	BREQ	$R_d \neq R_r$	$Z = 0$	BRNE	Unsigned
$R_d \leq R_r$	$C + Z = 1$	BRSH <sup>(1)</sup>	$R_d > R_r$	$C + Z = 0$	BRLO	Unsigned
$R_d < R_r$	$C = 1$	BRLO/BRCS	$R_d \geq R_r$	$C = 0$	BRSH/BRCC	Unsigned
Carry	$C = 1$	BRCS	No carry	$C = 0$	BRCC	Simple
Negative	$N = 1$	BRMI	Positive	$N = 0$	BRPL	Simple
Overflow	$V = 1$	BRVS	No overflow	$V = 0$	BRVC	Simple
Zero	$Z = 1$	BREQ	Not zero	$Z = 0$	BRNE	Simple

Note: 1. Interchange Rd and Rr in the operation before the test, i.e., CP Rd,Rr → CP Rr,Rd

- Reference: (doc0856) [AVR Instruction Set Manual](#)

— Page 10

CprE 288, ISU, Fall 2011

47

## Control Instruction Examples: IF

**R17 R2**  
if (X == Y) X = X+Y;

CP R17, R2 ; compare X & Y

BRNE SKIPADD ; Z=1 iff X==Y

ADD R17, R2 ; X = X+Y

SKIPADD: ADD R1, R2 ; do something

CprE 288, ISU, Fall 2011

48



## Control Instruction Examples: IF/Else

```

R17 R2
if (X == Y) X = X+Y;
else X = X-Y;

CP R17, R2      ; compare X & Y
BREQ THENPART  ; Z=1 iff X==Y
SUB R17, R2; X = X-Y : ELSEPART
JMP SKIPTHENPART ;do not do THENPART
                  ;as well
THENPART: ADD R17, R2      ; X=X+Y
SKIPTHENPART:

```

CprE 288, ISU, Fall 2011

49

## While/For LOOP

```

R17      R2 R1
for (i=0; i < 100; i++) X = X+Y;

CLR R17      ;i=0
LOOP: ADD R2, R1 ; X=X+Y
INC R17      ;i=i+1
CPI R17, 100 ; i-100
BRLT LOOP    ; branch on: N flag = 1

```

CprE 288, ISU, Fall 2011

50

## While/For LOOP: Can we do better?

```

R17      R2 R1
for (i=0; i < 100; i++) X = X+Y;

LDI R17, 100 ; i = 100
LOOP: ADD R2, R1 ; X=X+Y
DEC R17      ;i=i-1
BRNE LOOP    ; branch on not zero

```

CprE 288, ISU, Fall 2011

51

## Processing an Array

```

char A[100]; // Assume address is at 0x0200 ( X )
for (i=0; i < 100; i++) A[i] = A[i]+1;
R17      R16

; X is R27:R26
LDI R26, 0x00 ;XL holds LB of address A
LDI R27, 0x02 ;XH holds MB of address A
LDI R17, 100 ; i = 100
LOOP: LD R16, X ; A[i]
INC R16 ; A[i]=A[i]+1
ST X, R16
ADIW R26, 1 ; next A[i]
DEC R17 ;i=i-1
BRNE LOOP ; branch on not zero

```

CprE 288, ISU, Fall 2011

52

## Processing an Array

```

char A[100]; // Assume address is at 0x0200 ( X )
for (i=0; i < 100; i++) A[i] = A[i]+1;
R17      R16

; X is R27:R26
LDI R26, 0x00 ;XL holds LB of address A
LDI R27, 0x02 ;XH holds MB of address A
LDI R17, 100 ; i = 100
LOOP: LD R16, X ; A[i]
INC R16 ; A[i]=A[i]+1
ST X+, R16
ADIW R26, 1 ; next A[i]
DEC R17 ;i=i-1
BRNE LOOP ; branch on not zero

```

CprE 288, ISU, Fall 2011

53

## Example: String Copy

```

void strcpy (char *dst, char *src)
{
    char ch;

    do {
        ch = *src++;
        *dst++ = ch;
    } while (ch);
}

```

CprE 288, ISU, Fall 2011

54

## Example: String Copy

```
; dst in R25:R24, src in R23:R22
strcpy:
    movw r30, r24      ; Z<=dst
    movw r26, r22      ; X<=src
loop:
    ld    r20, X+       ; ch=*src++
    st    Z+, r20       ; *dst++=ch
    tst   r20           ; ch==0?
    brne loop          ; loop if not
    ret
```

CprE 288, ISU, Fall 2011

55